Homework 1: Equivalence principle at work: charge in a lab

1. Uniformly accelerated charge

The trajectory of a uniformly accelerated charge in Minkowski spacetime is given by the following plots.

In[∘]:=

```
ParametricPlot[Evaluate@Table[{1/g Cosh[g τ], 1/g Sinh[g τ]}, {g, {0.35, 0.5, 1, 2, 14}}],
 {τ, -1, 1}, PlotRange → {{0, 3}, {-1, 1}},
 PlotLegends → LineLegend[{0.35, 0.5, 1, 2, 14}, LegendLabel → g], AxesLabel → {x, t}]
```

Out[∘]=



Let us now fix the value of the four acceleration to be some suitable parameter

In[∘]:=

```
g = 2
```

Out[∘]=

2

Then the solutions of constant X are

In[∘]:=

```
ParametricPlot[
 Evaluate@Table[{(1/g + X) Cosh[g T], (1/g + X) Sinh[g T]}, {X, {0.4, 1, 2, 5, 8}}],
 {T, -1, 1}, PlotRange → {{0, 10}, {-3, 3}},
 PlotLegends → LineLegend[{0.4, 1, 2, 5, 8}, LegendLabel → X], AxesLabel → {x, t}]
```

Out[∘]=



while those of constant T are

In[∘]:=
```
ParametricPlot[Evaluate@
  Table[{(1/g + X) Cosh[g T], (1/g + X) Sinh[g T]}, {T, {-0.1, 0.01, 0.1, 0.2, 0.5, 1, 2}}],
 {X, -2, 2}, PlotRange → {{-1, 1}, {-1, 1}}, PlotLegends →
  LineLegend[{-0.1, 0.01, 0.1, 0.2, 0.5, 1, 2}, LegendLabel → T], AxesLabel → {x, t}]
```

Out[∘]=



2. Field of a uniformly accelerated charge

We begin by defining our coordinates and the value of the retarded time with the corresponding position of the particle.

In[∘]:=

$$\text{coord} = \{t, x, \rho, \phi\};$$
$$\delta = \rho^2 + x^2 + L^2 - t^2;$$
$$\xi = \sqrt{\frac{\left(L^2 + t^2 - \rho^2 - x^2\right)^2}{4} + L^2 \rho^2};$$
$$tQ = \frac{t\,\delta - 2\,x\,\xi}{2\left(x^2 - t^2\right)};$$
$$xQ = \frac{x\,\delta - 2\,t\,\xi}{2\left(x^2 - t^2\right)};$$

We can now create our four potential

In[∘]:=

$$A = \frac{Q}{t\,xQ - x\,tQ}\{-xQ, tQ, 0, 0\};$$

Our field strength tensor can now be defined

In[∘]:=

```
F = Table[
   D[A[[n]], coord[[m]]] - D[A[[m]], coord[[n]]] // FullSimplify, {m, 1, 4}, {n, 1, 4}];
F // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} 0 & \frac{4 L^2 \left(L^2+t^2-x^2+\rho^2\right)}{\left(L^4+2 L^2 \left(t^2-x^2+\rho^2\right)+\left(-t^2+x^2+\rho^2\right)^2\right)^{3/2}} & -\frac{8 L^2 x \rho}{\left(L^4+2 L^2 \left(t^2-x^2+\rho^2\right)+\left(-t^2+x^2+\rho^2\right)^2\right)^{3/2}} & 0 \\ -\frac{4 L^2 \left(L^2+t^2-x^2+\rho^2\right)}{\left(L^4+2 L^2 \left(t^2-x^2+\rho^2\right)+\left(-t^2+x^2+\rho^2\right)^2\right)^{3/2}} & 0 & \frac{8 L^2 t \rho}{\left(L^4+2 L^2 \left(t^2-x^2+\rho^2\right)+\left(-t^2+x^2+\rho^2\right)^2\right)^{3/2}} & 0 \\ \frac{8 L^2 x \rho}{\left(L^4+2 L^2 \left(t^2-x^2+\rho^2\right)+\left(-t^2+x^2+\rho^2\right)^2\right)^{3/2}} & -\frac{8 L^2 t \rho}{\left(L^4+2 L^2 \left(t^2-x^2+\rho^2\right)+\left(-t^2+x^2+\rho^2\right)^2\right)^{3/2}} & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

3. Sitting on the charge

In[∘]:=

```
Clear[g]
```

The modified Coulomb potential is

In[∘]:=

$$r = \sqrt{X^2 + \rho^2};$$

$$\Phi = \frac{Q}{r}\,\frac{1 + g\,X + \frac{g^2 r^2}{2}}{\sqrt{1 + g\,X + \frac{g^2 r^2}{4}}};$$

$$\varphi := \frac{\Phi}{1 + g\,X};$$

We normalize by the charge.

In[∘]:=

```
Q = 1
```

Out[∘]=

```
1
```

Then the contour plots of the potential are the following.

In[∘]:=

```
Table[ContourPlot[φ, {ρ, -3, 3}, {X, -1/g, 3}, PlotLegends → Automatic],
 {g, {0.1, 0.25, 0.3, 0.5, 1}}]
```

Out[ ]= {



1.2
1.0
0.8
0.6
0.4
2
0
–2
–4
–6
–8
–10
–3 –2 –1 0 1 2 3

,



2.50
2.25
2.00
1.75
1.50
1.25
1.00
0.75
0.50
0.25
3
2
1
0
–1
–2
–3
–4
–3 –2 –1 0 1 2 3

,



2.5
2.0
1.5
1.0
0.5
3
2
1
0
–1
–2
–3
–3 –2 –1 0 1 2 3

,



3.5
3.0
3



4.0
3.5
3.0
3



We can now define our new coordinates and the metric, potential and field strength in them.

In[ ]:=
```
coordp = {T, X, ρ, ϕ};
gp = {{-(1 + g X)^2, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, 1, 0}, {0, 0, 0, ρ^2}};
Ap = {-Φ, 0, 0, 0};
Fp = Table[D[Ap[[n]], coordp[[m]]] - D[Ap[[m]], coordp[[n]]] // FullSimplify,
   {m, 1, 4}, {n, 1, 4}];
```

Then our field strength is.

In[ ]:=
```
Fp // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 0 & -\frac{4\,(1+g\,X)\,\left(X\,(2+g\,X)-g\,\rho^2\right)}{\left(X^2+\rho^2\right)^{3/2}\,\left((2+g\,X)^2+g^2\,\rho^2\right)^{3/2}} & -\frac{8\,(1+g\,X)^2\,\rho}{\left(X^2+\rho^2\right)^{3/2}\,\left((2+g\,X)^2+g^2\,\rho^2\right)^{3/2}} & 0 \\ \frac{4\,(1+g\,X)\,\left(X\,(2+g\,X)-g\,\rho^2\right)}{\left(X^2+\rho^2\right)^{3/2}\,\left((2+g\,X)^2+g^2\,\rho^2\right)^{3/2}} & 0 & 0 & 0 \\ \frac{8\,(1+g\,X)^2\,\rho}{\left(X^2+\rho^2\right)^{3/2}\,\left((2+g\,X)^2+g^2\,\rho^2\right)^{3/2}} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

We can quickly check that this indeed satisfies Maxwell' s equations.

In[ ]:=
```
Table[∑_{m=1}^{4} D[√(-Det[gp]) (Inverse[gp].Fp.Inverse[gp])[[m, n]], coordp[[m]]] // Simplify,
 {n, 1, 4}]
```

Out[ ]=
```
{0, 0, 0, 0}
```